Pensieve header: The double and meta-double of the 2D pencil; continues pensieve://2017-05/.

Issues:

1. S does not invert R. (Perhaps because H must be interpreted as $e^{\hbar h}$).
2. dm is not meta-associative.
3. R doesn't satisfy YB.
4. Improve DUForm.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-06"];
```

The "degree carrier" is ħ, and all "coupling constants" are proportional to it.

```
$TD = 3; ħ /: ħ^d_. /; d > $TD := 0;
```

The 2D Lie BiAlgebra Pencil

We hope to stick to $A = e^{\hbar\beta a}$ and to $B = e^{\hbar\alpha b}$, where $[a, x] = \alpha x$ and $[b, y] = -\beta y$.
Also, $\Delta_{12}(a, A, x, b, B, y) = (a_1 + a_2,\ A_1 A_2,\ x_1 + A_1 x_2,\ b_1 + b_2,\ B_1 B_2,\ y_1 B_2 + y_2)$.
Also, (a, x) and $\hbar(b, y)$ are dual bases.

```
AlgebraAtom = a | A[_] | x | b | B[_] | y;
$PBWRule = {A[_] → 1, a → 2, x → 3, y → 4, B[_] → 5, b → 6};
```

```
B[a, x]  = α x; B[x, A[n_]] = (e^(n ħ α β) - 1) U[A[n], x]; B[a, A[_]] = 0;
B[y, b]  = β y; B[B[n_], y] = (e^(n ħ α β) - 1) U[y, B[n]]; B[b, B[_]] = 0;
Δ[a] = U1[a] U2[] + U1[] U2[a]; Δ[A[n_]] := U1[A[n]] U2[A[n]];
Δ[x] = U1[x] U2[] + U1[A[1]] U2[x];
Δ[b] = U1[b] U2[] + U1[] U2[b]; Δ[B[n_]] := U1[B[n]] U2[B[n]];
Δ[y] = U1[y] U2[B[1]] + U1[] U2[y];
S[a] = -a; S[A[n_]] := A[-n]; S[x]  = - U[A[-1], x];
S[b] = -b; S[B[n_]] := B[-n]; S[y]  = - U[y, B[-1]];
Si[a] = -a; Si[A[n_]] := A[-n]; Si[x]  = - U[x, A[-1]];
Si[b] = -b; Si[B[n_]] := B[-n]; Si[y]  = - U[B[-1], y];
(* This extra line is annoying *)
```

```
ExpandAB[ℰ_] := Expand@Normal@Series[ℰ //. {
      c_. U_i_[λ___, A[n_], ρ___] ⧴
       Expand[c Sum[((-1)^d ħ^d β^d n^d)/d! U_i[λ, Sequence @@ Table[a, {d}], ρ], {d, 0, $TD}]],
      c_. U_i_[λ___, B[n_], ρ___] ⧴ Expand[
        c Sum[((-1)^d ħ^d α^d n^d)/d! U_i[λ, Sequence @@ Table[b, {d}], ρ], {d, 0, $TD}]]
     },
    {ħ,
     0,
     $TD}]
```

UEA with provisional modification

This section is based on penseive://Projects/UEA/.

```
B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : AlgebraAtom, y_] := Expand[c B[x, y]];
B[y_, c_ * x : AlgebraAtom] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
B[x_, x_] = 0;
B[y_, x_] := Expand[-B[x, y]];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. $PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. $PBWRule];
UU_i_[] := U_i[]; UU_i_[1] := U_i[];
UU_i_[x_[n_]^p_] := U_i[x[n p]];
UU_i_[x_^p_] := UU_i @@ Table[x, {p}];
UU_i_[ℰ_] := ℰ /. {
    U[xs__] :→ UU_i[xs],
    x : AlgebraAtom :→ U_i[x]
   };
UU_i_[x_, xs__] := UU_t1[x] UU_t2[xs] // Expand // m_{t1,t2→i};
USimp[ℰ_] := Collect[ℰ, Times[U_[__] ..], Expand];
USimp[ℰ_] := Expand[ℰ];
```

```
m_s__[0] = 0;
m_s__[x_Plus] := m_s /@ x;
m_{i_→j_}[ℰ_] := ℰ /. U_i → U_j;
```

```
m_{i_,j_→k_}[c_. U_i_[x___] U_j_[]] := c U_k[x];
m_{i_,j_→k_}[c_. U_i_[] U_j_[y___]] := c U_k[y];
m_{i_,j_→k_}[c_. U_i_[xx___, x_[n1_]] U_j_[x_[n2_], yy___]] :=
  USimp[c If[n1 + n2 == 0, U_i[xx] U_j[yy], U_i[xx, x[n1 + n2]] U_j[yy]] // m_{i,j→k}];
m_{i_,j_→k_}[c_. U_i_[xx___, x_] U_j_[y_, yy___]] := If[x ≤ y,
   c U_k[xx, x, y, yy],
   ((U_i[xx] (U_j[y, x] + UU_j[B[x, y]]) // Expand // m_{i,j→i}) U_j[yy] // Expand // m_{i,j→k})
    c // USimp
  ];
```

```
Supp[ℰ_] := Union@Cases[{ℰ}, U_i_[___] :→ i, ∞];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
x_ ** y_ := Module[
   {Sx = Supp[x], Sy = Supp[y], is, σ, z},
   If[MatchQ[Sx ⋃ Sy, {_Integer ...}] && Min[Sx ⋃ Sy] < 0,
    is = Abs[Sx] ⋂ Abs[Sy];
    z = x; Do[z = m_{-i→-σ@i}[m_{i→σ@i}[z]], {i, is}];
    z = USimp[y z]; Do[z = dm_{σ@i,i→i}[z], {i, is}];
    z,
    (* else *) is = Sx ⋂ Sy;
    z = x; Do[z = m_{i→σ@i}[z], {i, is}];
    z = USimp[y z]; Do[z = m_{σ@i,i→i}[z], {i, is}];
    z
   ]
  ];
UB[x_, y_] := USimp[x ** y - y ** x];
```

```
O[func_] := Normal@Series[func, {ℏ, 0, $TD}];
O[specs_, func_] := Module[{rules, vars, elems},
  rules = Union @@ Cases[{specs}, U_[u___] :> Cases[{u}, r_Rule], ∞];
  vars = First /@ rules; elems = Last /@ rules;
  USimp@Total[CoefficientRules[O[func], vars] /. (ps_ → c_) :> c (
       specs /. MapThread[((#1 → _) :> #3^#2) &, {vars, ps, elems}] /. U_i_ :> UU_i
      )]
 ]
```

The 2D Lie BiAlgebra Pencil, Testing

```
O[U_1[a → a], e^(ℏ β a)]
```

$U_1[\,] + \beta\hbar\, U_1[a] + \frac{1}{2}\beta^2\hbar^2\, U_1[a, a] + \frac{1}{6}\beta^3\hbar^3\, U_1[a, a, a]$

```
USimp@With[{An = O[U_1[a → a], e^(-n ℏ β a)]}, UB[U_1[x], An] - O[e^(n ℏ α β) - 1] An ** U_1[x]]
```

```
0
```

```
B[x, A[3]]
```

$\left(-1 + e^{3\alpha\beta\hbar}\right) U[A[3], x]$

```
$TD = 6;
USimp@With[{Bn = O[U_1[b → b], e^(-n ℏ α b)]}, UB[Bn, U_1[y]] - O[e^(n ℏ α β) - 1] U_1[y] ** Bn]
```

```
0
```

```
z = U_1[a, A[2], x, x, x] U_2[a, a, x] U_3[a, a, A[-3], x];
(z // m_{1,2→1} // m_{1,3→1}) - (z // m_{2,3→2} // m_{1,2→1})
```

```
0
```

```
z = U1[y, y, y, b, B[2]] U2[y, b, b] U3[y, b, b, B[-3]];
(z // m1,2→1 // m1,3→1) - (z // m2,3→2 // m1,2→1)
```

```
0
```

The Co-Product and Co-Associativity

```
Δi_→j_,k_[ℰ_] := USimp@Module[{tj, tk}, ℰ /. {
      Ui[] → Uj[] Uk[],
      Ui[f_, fs___] :→
        (USimp[(Δ[f] /. {U1 → Uj, U2 → Uk}) Δi→tj,tk[Ui[fs]]] // mj,tj→j // mk,tk→k)
     }];
Δi_→j_,k_,l__[ℰ_] := ℰ // Δi→j,k // Δk→k,l
```

```
Δ1→1,2[U1[#]] & /@ {a, A[7], x, y, b, B[-3]}
```

```
{U1[a] U2[] + U1[] U2[a], U1[A[7]] U2[A[7]], U1[x] U2[] + U1[A[1]] U2[x],
 U1[] U2[y] + U1[y] U2[B[1]], U1[b] U2[] + U1[] U2[b], U1[B[-3]] U2[B[-3]]}
```

```
{lhs = U1[x] // Δ1→1,2 // Δ2→2,3, rhs = U1[x] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

```
{U1[x] U2[] U3[] + U1[A[1]] U2[x] U3[] + U1[A[1]] U2[A[1]] U3[x],
 U1[x] U2[] U3[] + U1[A[1]] U2[x] U3[] + U1[A[1]] U2[A[1]] U3[x], True}
```

```
U1[y] // Δ1→1,2
```

```
U1[] U2[y] + U1[y] U2[B[1]]
```

```
{lhs = U1[y] // Δ1→1,2 // Δ2→2,3, rhs = U1[y] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

```
{U1[] U2[] U3[y] + U1[] U2[y] U3[B[1]] + U1[y] U2[B[1]] U3[B[1]],
 U1[] U2[] U3[y] + U1[] U2[y] U3[B[1]] + U1[y] U2[B[1]] U3[B[1]], True}
```

```
z = U1[a, A[2], x, x, x] U2[a, a, A[-3], x];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

```
z = U1[y, y, y, b, B[2]] U2[y, b, b, B[-3]];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

The Antipode

```
Si_[ℰ_] := Module[{ti}, USimp[
    ℰ /. Ui[x_, xs___] :→ mti,i→i[Expand[UUi[S[x]] Sti[Uti[xs]]]]
   ]];
Sii_[ℰ_] := Module[{ti}, USimp[
    ℰ /. Ui[x_, xs___] :→ mti,i→i[Expand[UUi[Si[x]] Siti[Uti[xs]]]]
   ]];
```

```
{U1[x] // S1 // S1, U1[y] // S1 // S1}
```

$\{e^{\alpha\beta\hbar} U_1[x], e^{\alpha\beta\hbar} U_1[y]\}$

```
{U1[x] // S1 // Si1, U1[y] // S1 // Si1}
```

$\{U_1[x], U_1[y]\}$

```
{z = U1[]; (z // Δ1→1,2 // S1 // m1,2→1), z = U1[]; (z // Δ1→1,2 // S2 // m1,2→1)}
```

$\{U_1[], U_1[]\}$

```
z = U1[a, A[3], x, x];
{z // Δ1→1,2 // S2 // m1,2→1, z // Δ1→1,2 // S1 // m1,2→1}
```

{0, 0}

```
z = U1[y, y, y, b, B[-3]];
{z // Δ1→1,2 // S2 // m1,2→1, z // Δ1→1,2 // S1 // m1,2→1}
```

{0, 0}

```
{z = U1[a, A[2], x, x, x] U2[a, a, A[-3], x]; (z // m1,2→1 // S1) - (z // S1 // S2 // m2,1→1),
 z = U1[y, y, y, b, B[2]] U2[y, b, b, b, B[6]]; (z // m1,2→1 // S1) - (z // S1 // S2 // m2,1→1)}
```

{0, 0}

```
{z = U1[a, A[2], x, x, x]; (z // S1 // Δ1→2,1) - (z // Δ1→1,2 // S1 // S2),
 z = U1[y, y, y, b, B[2]]; (z // S1 // Δ1→2,1) - (z // Δ1→1,2 // S1 // S2)
}
```

{0, 0}

The Pairing

```
P[U[], U[B[_]]] = P[U[A[_]], U[]] = 1;
P[U[], U[__]] = P[U[__], U[]] = 0;
(    P[U[a], U[b]] = ℏ^-1         P[U[a], U[B[n_]]] = -n α       P[U[a], U[y]] = 0
 P[U[A[n_]], U[b]] = -n β  P[U[A[n_]], U[B[m_]]] = e^(n m ℏ α β)  P[U[A[_]], U[y]] = 0
     P[U[x], U[b]] = 0            P[U[x], U[B[_]]] = 0          P[U[x], U[y]] = ℏ^-1 );
```

```
P[U[], U[]] = 1;
Pi_,j_[ℰ_] := USimp[ℰ /. Ui[xs___] Uj[ys___] → P[U[xs], U[ys]]];
```

The pairing sequence: ⟨one,one⟩ (above), ⟨many,one⟩, ⟨many,many⟩.

```
P[U[x_, xs__], U[y_]] := P[U[x, xs], U[y]] =
   Module[{i, j, k, l}, USimp[Ui[x] UUj[xs] Δk→k,l[Uk[y]]] // Pi,k // Pj,l];
P[U[xs__], U[y_, ys__]] := P[U[xs], U[y, ys]] =
   Module[{i, j, k, l}, USimp[Δi→i,j[UUi[xs]] Uk[y] UUl[ys]] // Pi,k // Pj,l];
```

```
z = U_i[a] U_j[x] U_k[y];
{m_{i,j→i}[z] - m_{j,i→i}[z], Δ_{k→k,l}[z] - Δ_{k→l,k}[z]}
```

$\{\alpha U_i[x] U_k[y], -U_i[a] U_j[x] U_k[y] U_l[] + U_i[a] U_j[x] U_k[] U_l[y] - U_i[a] U_j[x] U_k[B[1]] U_l[y] + U_i[a] U_j[x] U_k[y] U_l[B[1]]\}$

```
Table[z = U_i[xi] U_j[xj] U_k[yk];
 {(m_{i,j→i}[z] - m_{j,i→i}[z]) // P_{i,k}, (Δ_{k→k,l}[z] - Δ_{k→l,k}[z]) // P_{i,k} // P_{j,l}},
 {xi, {a, x}}, {xj, {a, x}}, {yk, {b, y}}]
```

$\{\{\{\{0,0\},\{0,0\}\},\{\{0,0\},\{\frac{\alpha}{\hbar},\frac{\alpha}{\hbar}\}\}\},\{\{\{0,0\},\{-\frac{\alpha}{\hbar},-\frac{\alpha}{\hbar}\}\},\{\{0,0\},\{0,0\}\}\}\}$

```
Table[z = U_i[xi] U_k[yk] U_l[yl];
 {(Δ_{i→i,j}[z] - Δ_{i→j,i}[z]) // P_{i,k} // P_{j,l}, (m_{k,l→k}[z] - m_{l,k→k}[z]) // P_{i,k}},
 {xi, {a, x}}, {yk, {b, y}}, {yl, {b, y}}]
```

$\{\{\{\{0,0\},\{0,0\}\},\{\{0,0\},\{0,0\}\}\},\{\{\{0,0\},\{-\frac{\beta}{\hbar},-\frac{\beta}{\hbar}\}\},\{\{\frac{\beta}{\hbar},\frac{\beta}{\hbar}\},\{0,0\}\}\}\}$

```
lhs = Factor@Table[ħ^n P[U @@ Table[x, {n}], U @@ Table[y, {n}]], {n, $TD = 7}]
```

$\{1, 1+e^{\alpha\beta\hbar}, (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}), (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$
$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\}$

```
rhs = Simplify@FunctionExpand@Table[QFactorial[n, e^(ħ α β)], {n, $TD = 7}]
```

$\{1, 1+e^{\alpha\beta\hbar}, (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}), (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$
$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$
$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\}$

```
MapThread[Equal, {lhs, rhs}]
```

{True, True, True, True, True, True, True}

```
P[U[a, a, a, a, a], U[b, b, b, b, b]]
```

$\frac{120}{\hbar^5}$

```
Table[P[z1, z2],
  {z1, {U[], U[a], U[x], U[a, a], U[a, x], U[x, x],
    U[a, a, a], U[a, a, x], U[a, x, x], U[x, x, x]}}, {z2, {U[], U[b], U[y],
    U[b, b], U[y, b], U[y, y], U[b, b, b], U[y, b, b], U[y, y, b], U[y, y, y]}}
] //
 MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{\hbar} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{\hbar} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{2}{\hbar^2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{\hbar^2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^2}+\frac{e^{\alpha\beta\hbar}}{\hbar^2} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{6}{\hbar^3} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{2}{\hbar^3} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^3}+\frac{e^{\alpha\beta\hbar}}{\hbar^3} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^3}+\frac{2e^{\alpha\beta\hbar}}{\hbar^3}+\frac{2e^{2\alpha\beta\hbar}}{\hbar^3}+\frac{e^{3\alpha\beta\hbar}}{\hbar^3} \end{pmatrix}$$

The Double

```
DUForm[ℰ_] := ℰ //. U_i_[fs___] U_j_[gs___] /; i + j == 0 ∧ j > 0 :> "DU"_j[fs, gs];
(DU_i_[a] := U_-i[] U_i[a]; DU_i_[x] := U_-i[] U_i[x];
 DU_i_[y] := U_-i[y] U_i[]; DU_i_[b] := U_-i[b] U_i[]; )
```

```
DU_1 /@ {a, x, y, b}
```

$\{U_{-1}[\,]\,U_1[a],\ U_{-1}[\,]\,U_1[x],\ U_{-1}[y]\,U_1[\,],\ U_{-1}[b]\,U_1[\,]\}$

```
dm_i_,j_→k_[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
   ℰ // Δ_-j→t1,t2,t3 // S_t3 // Δ_i→h1,h2,h3 // P_h1,t1 // P_h3,t3 // m_j,h2→k // m_-i,t2→-k];
altdm_i_,j_→k_[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
   ℰ // Δ_i→h1,h2,h3 // Si_h3 // Δ_-j→t1,t2,t3 // P_h1,t1 // P_h3,t3 // m_j,h2→k // m_-i,t2→-k];
dΔ_i_→j_,k_[ℰ_] := ℰ // Δ_i→j,k // Δ_-i→-j,-k;
dS_i_[ℰ_] := Module[{h}, (ℰ // m_i→h // Si_h // S_-i) U_-h[] U_i[] // Expand // dm_h,i→i];
```

```
DU_1[x] DU_2[a] // dm_1,2→1 // DUForm
```

$DU_1[a, x]$

```
U_-1[] U_1[a] U_-2[] U_2[x] // dm_1,2→1 // DUForm
```

$-\alpha\, DU_1[x] + DU_1[a, x]$

```
U_-1[] U_1[a] U_-2[b] U_2[] // dm_1,2→1 // DUForm
```

$DU_1[b, a]$

```
{U₋₁[] U₁[a] U₋₂[y] U₂[] // dm_{2,1→1}, U₋₁[] U₁[a] U₋₂[y] U₂[] // dm_{1,2→1}} // DUForm
```

$\{DU_1[y, a], \alpha\, DU_1[y] + DU_1[y, a]\}$

```
(U₋₁[] U₁[x] U₋₂[b] U₂[] - U₋₁[b] U₁[] U₋₂[] U₂[x]) // dm_{1,2→1} // DUForm
```

$-\beta\, DU_1[x]$

```
U₋₁[] U₁[A[1]] U₋₂[y] U₂[] // dm_{1,2→1}
```

$e^{-\alpha\beta\hbar}\, U_{-1}[y]\, U_1[A[1]]$

```
U₋₁[] U₁[x] U₋₂[b] U₂[] // dm_{1,2→1}
```

$-\beta\, U_{-1}[\,]\, U_1[x] + U_{-1}[b]\, U_1[x]$

```
U₋₁[] U₁[x] U₋₂[B[1]] U₂[] // dm_{1,2→1}
```

$e^{\alpha\beta\hbar}\, U_{-1}[B[1]]\, U_1[x]$

```
DU₁[x] DU₂[y] // dm_{1,2→1} // DUForm
```

$$-\frac{DU_1[A[1]]}{\hbar} + \frac{DU_1[B[1]]}{\hbar} + DU_1[y, x]$$

```
z = U₋₁[] U₁[x] U₋₂[y] U₂[] U₋₃[b] U₃[];
(z // dm_{1,2→1} // dm_{1,3→1}) - (z // dm_{2,3→2} // dm_{1,2→1}) // DUForm
```

0

```
$TD = 5;
z = U₋₁[y, b, b] U₁[A[2], a, x, x] U₋₂[B[-1], y, y, b, b] U₂[a] U₋₃[y, y, b] U₃[a, a, x, x, x];
(z // dm_{1,2→1} // dm_{1,3→1}) - (z // dm_{2,3→2} // dm_{1,2→1}) // DUForm
```

0

```
(# → DUForm[dS₁[DU₁[#]]]) & /@ {a, x, y, b}
```

$\{a \to -DU_1[a],\ x \to -e^{-\alpha\beta\hbar}\, DU_1[A[-1], x],\ y \to -DU_1[y, B[-1]],\ b \to -DU_1[b]\}$

```
(DU₁[x] DU₂[y] // dm_{1,2→1} // dS₁) - (DU₁[x] DU₂[y] // dS₁ // dS₂ // dm_{2,1→1}) // ExpandAB
```

0

The R-Matrix

Quesne's formulas:

$$q = e^{\hbar\alpha\beta};\ \mathbb{e}_{q\_}[x\_] := \text{Exp}\left[\sum_{k=1}^{\$TD} \frac{(1-q)^k}{k\,(1-q^k)}\, x^k\right];$$

```
Table[Together@SeriesCoefficient[e_ρ[x], {x, 0, n}], {n, 0, $TD}]
```

$$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)},\right.$$
$$\left. 1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
R_{i_,j_} := 𝕆[U_{-i}[y → y, b → b] U_i[] U_{-j}[] U_j[a → a, x → x], 𝕖^{ℏ b a} 𝕖_q[ℏ x y]]
```

```
$TD = 3; R_{1,2} // DUForm
```

$DU_1[]\,DU_2[] + \hbar\,DU_1[b]\,DU_2[a] + \hbar\,DU_1[y]\,DU_2[x] + \frac{1}{2}\hbar^2\,DU_1[b,b]\,DU_2[a,a] + \hbar^2\,DU_1[y,b]\,DU_2[a,x] + \frac{1}{2}\hbar^2\,DU_1[y,y]\,DU_2[x,x] - \frac{1}{4}\alpha\beta\hbar^3\,DU_1[y,y]\,DU_2[x,x] + \frac{1}{6}\hbar^3\,DU_1[b,b,b]\,DU_2[a,a,a] + \frac{1}{2}\hbar^3\,DU_1[y,b,b]\,DU_2[a,a,x] + \frac{1}{2}\hbar^3\,DU_1[y,y,b]\,DU_2[a,x,x] + \frac{1}{6}\hbar^3\,DU_1[y,y,y]\,DU_2[x,x,x]$

```
$TD = 5; {lhs = R_{1,3} // dΔ_{1→1,2}, rhs = R_{2,3} ** R_{1,3}, lhs - rhs} // Last // ExpandAB // DUForm
```

0

```
$TD = 5; {lhs = R_{1,2} // dΔ_{2→2,3}, rhs = R_{1,2} ** R_{1,3}, lhs - rhs} // Last // ExpandAB // DUForm
```

0

```
$TD = 5; Table[
 {lhs = R_{1,2} ** dΔ_{1→2,1}[f], rhs = dΔ_{1→1,2}[f] ** R_{1,2}, lhs - rhs} // Last // ExpandAB // DUForm,
 {f, {U_{-1}[] U_1[a], U_{-1}[] U_1[x], U_{-1}[b] U_1[], U_{-1}[y] U_1[]}}]
```

{0, 0, 0, 0}

```
$TD = 5;
{lhs = ExpandAB[R_{1,2} ** R_{1,3} ** R_{2,3}],
  rhs = ExpandAB[R_{2,3} ** R_{1,3} ** R_{1,2}], Coefficient[lhs - rhs, ℏ^$TD]} // Last
```

0

```
$TD = 5; Si_2[R_{1,2}] ** R_{1,2} // ExpandAB // DUForm
```

$DU_1[]\,DU_2[]$

Cuaps (Unfinished)

Drinfeld element u (probably the wrong S is used, need to implement dS first.

```
u_{i_} := R_{1,2} // S_2 // m_{2,1→i}
```

```
$TD = 1; u_1 // DUForm
```

$DU_1[]\,U_{-2}[] - \hbar\,DU_1[b,a]\,U_{-2}[] - \hbar\,DU_1[y, A[-1], x]\,U_{-2}[]$

```
$TD = 1; Table[
 {lhs = f ** u_1, rhs = u_1 ** S_1[S_1[f]], lhs - rhs} // Last // ExpandAB // DUForm,
 {f, {U_{-1}[] U_1[a], U_{-1}[] U_1[x], U_{-1}[b] U_1[], U_{-1}[y] U_1[]}}]
```

$\{0, 0, 0, -\alpha\beta\hbar\,DU_1[y]\,U_{-2}[]\}$

The Central Element

```
c1 = α U-1[b] U1[] + β U-1[] U1[a];
UB[c1, #] & /@ {U-1[] U1[a], U-1[] U1[x], U-1[y] U1[], U-1[b] U1[]}
{0, 0, 0, 0}
```

Commuting Exponentials

Commuting e^a with e^x:

```
$TD = 5; 𝕆[U1[x → x, a → a], e^(ℏ (μ x+ν a))] == 𝕆[U1[a → a, x → x], e^(ℏ (μ e^(-ℏ α ν) x+ν a))]
True
```

Commuting e^b with e^y:

```
$TD = 5; 𝕆[U1[b → b, y → y], e^(ℏ (μ y+ν b))] == 𝕆[U1[y → y, b → b], e^(ℏ (μ e^(-ℏ β ν) y+ν b))]
True
```

The co-product of e^a:

```
$TD = 5; Δ1→1,2[𝕆[U1[a → a], e^(ℏ μ a)]] == 𝕆[U1[a1 → a] U2[a2 → a], e^(ℏ μ a1) e^(ℏ μ a2)]
True
```

The co-product of e_q^x:

```
$TD = 5 ; (𝕆[U1[x → x], eq[ℏ μ x]] // Δ1→1,2 // ExpandAB) ==
 𝕆[U1[a1 → a, x1 → x] U2[x2 → x], eq[ℏ μ e^(-ℏ β a1) x2] eq[ℏ μ x1]]
True
```

The triple co-product of e_q^x:

```
$TD = 5; (𝕆[U1[x → x], eq[ℏ μ x]] // Δ1→1,2,3 // ExpandAB) -
 𝕆[U1[a1 → a, x1 → x] U2[a2 → a, x2 → x] U3[x3 → x],
  eq[ℏ μ e^(-ℏ β (a1+a2)) x3] eq[ℏ μ e^(-ℏ β a1) x2] eq[ℏ μ x1]]
0
```

The co-product of e_q^y:

```
$TD = 5 ; (𝕆[U1[y → y], eq[ℏ ν y]] // Δ1→1,2 // ExpandAB) ==
 𝕆[U1[y1 → y] U2[y2 → y, b2 → b], eq[ℏ ν y2] eq[ℏ ν e^(-ℏ α b2) y1]]
True
```

The triple co-product of e_q^y:

```
$TD = 5 ; (𝕆[U1[y → y], eq[ℏ ν y]] // Δ1→1,2,3 // ExpandAB) ==
 𝕆[U1[y1 → y] U2[y2 → y, b2 → b] U3[y3 → y, b3 → b],
  eq[ℏ ν y3] eq[ℏ ν e^(-ℏ α b3) y2] eq[ℏ ν e^(-ℏ α (b2+b3)) y1]]
True
```

The inverse-antipode on $\mathbb{e}_q^x$:

```
$TD = 8;
(𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Si₁ // ExpandAB) -
 (𝕆[U₁[a → a, x → x], Sum[((-ℏ μ)^k q^(-k (k+1)/2) 𝕖^(ℏ β k a) x^k)/QFactorial[k, q], {k, 0, $TD}]])
```

```
0
```

Pairing $\mathbb{e}_q^x$ with $\mathbb{e}_q^y$:

```
$TD = 5;
$TD *= 2;
lhs = 𝕆[U₋₁[y → y], 𝕖_q[ℏ ν y]] 𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Expand // P₁,₋₁ // ExpandAB;
($TD /= 2; 𝕖_q[ℏ μ ν] - lhs // ExpandAB)
```

```
0
```

Pairing $S^{-1}\,\mathbb{e}_q^x$ with $\mathbb{e}_q^y$: